lab3ans

Calculus A, Lab 3 Solutions

Tim McLarnan

As suggested, we begin by defining our function:

```
f(x) = 60*x/(x^2+21)
f(x)
```

$\frac{60\,x}{x^2+21}$

```
plot(f(x), (x,-15,15)).show(aspect_ratio=1)
```




```
plot(f(x), (x,-2,8)).show(aspect_ratio=1)
```



The slope of this curve at $x = 3$ looks to be somewhere around 1. One could try to estimate this number by plotting the function and sketching the tangent line. One could also just draw line through the point $(3,\ f(3))$ until one hit something that looked tangent. Here are a couple of efforts that look pretty close. Either of these lines by itself looks fairly tangent to the curve, but seeing them together makes a pretty persuasive case that the real tangent line must have slope somewhere between 0.6 and 1.0. Maybe it's something like $3/4$?

```
p1 = plot(f(x), (x,1,8), color = 'blue')
p2 = plot(f(3) + 0.6*(x-3), (x,1,8), color = 'red')
p3 = plot(f(3) + 1.0*(x-3), (x,1,8), color = 'red')
show(p1 + p2 + p3, aspect_ratio=1)
```



The slopes of the secant lines through the point at $(3, 6)$ and the points at $x = 4, 6, 10, 2$ and 0 are

```
(f(4)-f(3))/(4-3), ((f(4)-f(3))/(4-3)).n()
```

$\left(\frac{18}{37}, 0.486486486486487\right)$

```
(f(6)-f(3))/(6-3), ((f(6)-f(3))/(6-3)).n()
```

$\left(\frac{2}{19}, 0.105263157894737\right)$

```
(f(10)-f(3))/(10-3), ((f(10)-f(3))/(10-3)).n()
```

$\left(-\frac{18}{121}, -0.148760330578512\right)$

```
(f(2)-f(3))/(2-3), ((f(2)-f(3))/(2-3)).n()
```

$\left(\frac{6}{5}, 1.20000000000000\right)$

```
(f(0)-f(3))/(0-3), ((f(0)-f(3))/(0-3)).n()
```

$(2, 2.00000000000000)$

These lines are sketched below, along with our approximate tangent line. Obviously the line closest to the tangent line at $x = 3$ is either the one going through the point on the curve where $x = 4$ or the one going through the point on the curve where $x = 2$. These two lines are the ones through the points on the curve closest to the point at $x = 3$.

```
p1 = plot(f(x), (x,-2,15), color = 'blue')
p2 = plot(f(3) + 0.75*(x-3), (x,-2,15), color = 'red')
p3 = plot(f(3) + ((f(4)-f(3))/(4-3))*(x-3), (x,-2,15), color = 'green')
p4 = plot(f(3) + ((f(6)-f(3))/(6-3))*(x-3), (x,-2,15), color = 'green')
p5 = plot(f(3) + ((f(10)-f(3))/(10-3))*(x-3), (x,-2,15), color = 'green')
p6 = plot(f(3) + ((f(2)-f(3))/(2-3))*(x-3), (x,-2,15), color = 'green')
p7 = plot(f(3) + ((f(0)-f(3))/(0-3))*(x-3), (x,-2,15), color = 'green')
show(p1 + p2 + p3 + p4 + p5 + p6 + p7, ymax=9, aspect_ratio=1)
```



To get a decent numerical approximation to $f'(3)$, just compute difference quotients with the second point closer and closer to 3 until the first three digits of the reslt stop changing. The calculations below suggest that $f'(3)$ is close to 0.80000, which is in complete agreement with our initial guess of around 0.75.

```
(f(3.1)-f(3))/(3.1-3)
```

0.764456060111084

```
(f(3.01)-f(3))/(3.01-3)
```

0.796404536245738

```
(f(3.001)-f(3))/(3.001-3)
```

0.799640045336591

```
(f(3.0001)-f(3))/(3.0001-3)
```

0.799964000453545

```
(f(3.00001)-f(3))/(3.00001-3)
```

0.799996400052993

```
(f(2.99999)-f(3))/(2.99999-3)
```

0.800003600071352

I'll compute the slope analytically both by using *Sage* and just working by hand.

If we want to do it completely on our own, to see what happens and because we don't trust *Sage*, then we would do something like this: The slope of a secant line is

$$\frac{\frac{60x}{x^2+21}-6}{x-3}.$$

The derivative $f'(1)$ will therefore be

$$f'(3)=\lim_{x\to 3}\frac{\frac{60x}{x^2+21}-6}{x-3}.$$

To work out this limit, start by writing the difference quotient on two levels instead of 3 and putting the terms in the numerator over a common denominator. We get

$$\frac{1}{x-3}\left[\frac{60x}{x^2+21}-6\right]=\frac{1}{x-3}\left[\frac{60x-6x^2-126}{x^2+21}\right]=\frac{-6(x^2-10x+21)}{(x-3)(x^2+21)}=\frac{-6(x-3)(x-7)}{(x-3)(x^2+21)}=\frac{-6(x-7}{x^2+21}$$

The derivative is therefore

$$f'(3)=\lim_{x\to 3}\frac{\frac{60x}{x^2+21}-6}{x-3}=\lim_{x\to 3}\frac{-6(x-7)}{x^2+21}=\frac{24}{30}=\frac{4}{5}=0.8.$$

This agrees completely with our numerical approximation above.

This wasn't a difficult calculation; but if we wanted to avoid some of the algebra, we could just trust *Sage* to do it for us. Here's a *Sage*-assisted version of the calculation above:

```
(f(x)-f(3))/(x-3)
```

$$\frac{6\left(\frac{10\,x}{x^2+21}-1\right)}{x-3}$$

```
_.simplify_rational()
```

$$\frac{-6\,(x-7)}{x^2+21}$$

```
_(x=3)
```

$$\frac{4}{5}$$

Of course, *Sage* can also do it in a single command,

```
limit((f(x)-f(3))/(x-3), x=3)
```

$$\frac{4}{5}$$

and *Sage* has a built-in command for computing derivatives that could be used if one wanted a complete black-box solution:

```
f(x).diff()(x=3)
```

$$\frac{4}{5}$$

Since the slope of the tangent line is $4/5$, its equation will be $y-6=\frac{4}{5}(x-3)$, or

$$y = \frac{4}{5}x + \frac{18}{5}.$$

Plotting the original curve and this line together makes it look like we really do have the correct tangent:

```
p1 = plot(f(x), (x,-2,15), color = 'blue')
p2 = plot((4/5)*x+18/5, (x,-2,15), color = 'red')
show(p1 + p2, ymax=9, aspect_ratio=1)
```



I won't play further here with the final extra-credit problem. By this point, we have done many of those calculations in class.